

Eiffel SDL multimedia library (ESDL)  
API extensions  
SEMESTER THESIS

Benno Baumgartner

30th March 2004

<http://eiffelsdl.sf.net>

**Student:** Benno Baumgartner ([benno@student.ethz.ch](mailto:benno@student.ethz.ch))

**Student-No:** 98-727-589

**Supervising Assistant:** Till G. Bay

**Supervising Professor:** Bertrand Meyer

## Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
<b>2</b>	<b>Implementation of the Extensions</b>	<b>4</b>
2.1	Overview	4
2.2	ESDL_DRAWABLE	4
2.3	ESDL_SURFACE	9
2.3.1	Example	10
2.4	ESDL_SPRITE	10
2.4.1	ESDL_ANIMATION	10
2.4.2	ESDL_RANDOM_SPRITE	11
2.4.3	Example	11
2.5	ESDL_TILE_PATTERN	11
2.5.1	ESDL_TILE_PATTERN_HORIZONTAL	12
2.5.2	ESDL_TILE_PATTERN_VERTICAL	12
2.5.3	Example	12
2.6	ESDL_NEVER_ENDING_BACKGROUND	12
2.6.1	ESDL_NEVER_ENDING_BACKGROUND_HORIZONTAL	13
2.6.2	ESDL_NEVER_ENDING_BACKGROUND_VERTICAL	13
2.6.3	Example	14
2.7	ESDL_DRAWABLE_CONTAINER	14
2.7.1	Example	15
2.8	ESDL_STRING	16
2.8.1	ESDL_CHARACTER	16
2.8.2	ESDL_FONT	16
2.8.3	ESDL_BITMAPFONT	18
2.8.4	Example	19
2.9	ESDL_FRAME_COUNTER	19
2.9.1	Example	19
2.10	ESDL_TIME	19
2.10.1	Example	20
2.10.2	ESDL_TIME_SINGLETON	20
2.10.3	Callbacks	21
2.10.4	Example	21
2.10.5	ESDL_MUTEX	21
2.11	ESDL_COLLIDABLE	21
2.11.1	ESDL_RECT_COLLIDABLE	22
2.11.2	ESDL_COLLISION_DETECTOR	23
2.11.3	Example	23
<b>3</b>	<b>Examples</b>	<b>24</b>
3.1	Bitmap fonts	24
3.2	Drawable demo	25
3.3	Never ending background	27
3.4	Collidable string	28

3.5	ESDL Racing . . . . .	29
<b>4</b>	<b>Future Work</b>	<b>31</b>

## List of Figures

1	ESDL_DRAWABLE BON Diagram . . . . .	8
2	A never ending background . . . . .	13
3	A container . . . . .	15
5	Bitmap font Arial_256 . . . . .	18
4	ESDL_FONT BON Diagram . . . . .	18
6	Bitmap font example . . . . .	24
7	A drawable demo part 1 . . . . .	25
8	A drawable demo part 2 . . . . .	25
9	A demo of a never ending background . . . . .	27
10	A demo of collidable strings . . . . .	28
11	ESDL Racing level 1 . . . . .	29
12	ESDL Racing level 2 . . . . .	29
13	ESDL Racing level 3 . . . . .	30

## 1 Introduction

ESDL is a wrapper for SDL, the Simple Directmedia Layer library [4]. SDL is an open source C library that is very popular among Linux game developers. Besides Linux, SDL is also available for many other platforms such as Windows and MacOS. SDL is a very performing multimedia library composed of various subsystems for graphics, sound, networking, threading, CD-ROM access, window management, joystick handling, event handling and time management. The aim of this project is to provide the Eiffel community with a wrapper of SDL that has an easy to understand and memorable API, that is as performing as SDL is and that runs on different platforms. At the moment Windows and Linux are supported.

ESDL is being developed subsystem by subsystem. This semester thesis report describes the implementation of API extensions of ESDL 0.0.1. Among others I have extended the time subsystem, added extensions to the graphical subsystem that allows a developer to handle RGB bitmap graphics in a very flexible and easy way. I also added collision detection support.

These extensions allow to build complex games in a very easy way as the example application ESDL\_Racing in subsection 3.5 on page 29 demonstrates.

The graphical subsystem and the event loop are not described in this report but in the documentation of the first version of ESDL [1].

## 2 Implementation of the Extensions

### 2.1 Overview

In subsection 2.2 until 2.9 on page 19 all classes related to ESDL\_DRAWABLE are introduced. This classes provide powerful and very flexible functionalities to work with two dimensional RGB bitmaps in ESDL.

In subsections 2.10 on page 19 until 2.10.5 on page 21 all classes related to the time subsystem are covered. The time subsystem provides the programmer with functionalities related to time measurement that is used for all the animation support.

In subsection 2.11 on page 21 all classes related to collision detection are explained.

### 2.2 ESDL\_DRAWABLE

Every class in ESDL describing an object that can be drawn to an ESDL\_SURFACE should implement ESDL\_DRAWABLE. ESDL\_DRAWABLE is a deferred class. It's main features are *draw* and *draw\_part*:

#### deferred class

ESDL\_DRAWABLE

**feature** -- commands

```
draw (a_surface: ESDL_SURFACE) is
  -- Draws 'current' to 'a_surface'
  require
    a_surface_not_void: a_surface /= void
  deferred
  end

draw_part (rect: ESDL_RECT; a_surface: ESDL_SURFACE) is
  -- Draws rectangular part of 'current' defined by 'rect' to 'a_surface'
  require
    a_rect_not_void: rect /= void
    a_surface_not_void: a_surface /= void
  deferred
  end
```

**feature** -- status

```
width: INTEGER is
  -- The 'width' of 'current'
  -- The 'width' is the position of the right most pixel
  deferred
  ensure
    Result >= 0
  end

height: INTEGER is
  -- The 'height' of 'current'
  -- The 'height' is the position of the bottom most pixel
  deferred
  ensure
    Result >= 0
  end

x: INTEGER
  -- The distance of 'current' to the coordinate origin
  -- in 'x' (left -> right) direction

set_x (an_integer: INTEGER) is
  -- Sets 'x'
  require
```

```
    positive: an_integer >= 0
  do
    x := an_integer
  ensure
    set: x = an_integer
  end

y: INTEGER
  -- The distance of 'current' to the coordinate origin
  -- in 'y' (top -> down) direction

set_y (an_integer: INTEGER) is
  -- Sets 'y'
  require
    positive: an_integer >= 0
  do
    y := an_integer
  ensure
    set: y = an_integer
  end

set_x_y (an_x: INTEGER; an_y: INTEGER) is
  -- Sets the 'x' and 'y' value
  require
    positive: an_x >= 0 and an_y >= 0
  do
    x := an_x
    y := an_y
  ensure
    set: an_x = x and an_y = y
  end

invariant


  x_positive: x >= 0
  y_positive: y >= 0

end
```

The pixel with the coordinate (0,0) is always at the top left of *a\_surface* passed to *draw* and *draw\_part*. The coordinate (0,0) is called the origin. The *width* is defined as the distance between the origin and the right most pixel of the

drawable. The *height* is defined as the distance between the origin and the bottom most pixel of the drawable. This requires, that the *x* and *y* position is always positive. This limitation seems to be stronger then needed at first sight. But only that way the *width* and *height* of the ESDL\_DRAWABLE\_CONTAINER can be implemented consistent to the definition without extending the interface of ESDL\_DRAWABLE.

At the moment 11 effective classes inherit from ESDL\_DRAWABLE and can be used as drawables. See Figure 1 on the following page.



8



## 2.3 ESDL\_SURFACE

ESDL\_SURFACE is an ESDL\_DRAWABLE.

ESDL\_SURFACE is a two dimensional RGB bitmap. You can draw any ESDL\_DRAWABLE to an ESDL\_SURFACE. If you want to program in ESDL you first need a base surface: the screen. You can get a screen with the following sequence of code:

```
local
  esdl: ESDL
  screen: ESDL_SURFACE
do
  create esdl.make
  esdl.set_video_surface_width (width)
  esdl.set_video_surface_height (height)
  esdl.set_video_bpp (resolution)
  esdl.set_fullscreen (True)
  esdl.set_doublebuffered (True)
  esdl.set_hardware_surface (True)
  esdl.set_hwpalette (True)
  esdl.showcursor (False)
  esdl.initialize_video_surface
  screen := esdl.video_surface
end
```

Where *width* is the width of the screen and *height* is the height of the screen. You can use any combination as *width* and *height*, but you should use one that is directly supported by your graphics card, if you want to run your program in full screen mode. A supported resolution is in this case faster than an arbitrary combination. In Windows you can find out which resolutions your card supports by clicking right on the desktop and selecting Properties ▸ Settings. In the Screen area you can see all possible combinations. Very common ones are: 640 by 480 pixels, 800 by 600 pixels and 1024 by 768 pixels. Those should be supported by every modern video card.

The *resolution* is the number of bits used to represent color information. You can use 8, 16, 24 or 32 bits as *resolution*. You should avoid using a *resolution* of 24 bits because such a *resolution* is a lot slower than the others.

Once you have a screen you can draw other ESDL\_DRAWABLEs to it. You can also create your own ESDL\_SURFACES. There exist two possibilities for doing so:

1. *make\_from\_image* (*an\_image\_file\_name*: *STRING*)  
This feature loads an image from a file and creates an ESDL\_SURFACE

from this image. The format is the same as the format of the current screen. This feature uses the `SDL_image` library to load an image [6]. Supported file types are: BMP, PNM, XPM, XCF, PCX, GIF, JPG, TIF, PNG and LBM.

2. *make* (*w:INTEGER*; *h:INTEGER*)

This feature creates an `ESDL_SURFACE` with width *w* and height *h*. The format is the same as the format of the current screen.

You can give an `ESDL_SURFACE` a so called key color with the command *set\_colorkey*. Every pixel with the color defined with *set\_colorkey* is not drawn, but it will be transparent.

`ESDL_SURFACES` are collected by the Eiffel garbage collector. If you want to force that (at start of a new level for example) inherit from the class `MEMORY` and call *full\_collect*.

### 2.3.1 Example

```
local
  picture: ESDL_SURFACE
do
  create picture.make_from_image ("lovely.gif")
  picture.set_colorkey (255, 255, 255)
  picture.draw (screen)
  screen.redraw
end
```

## 2.4 ESDL\_SPRITE

An `ESDL_SPRITE` is an `ESDL_DRAWABLE`.

`ESDL_SPRITE` is a player for an `ARRAY` of `ESDL_DRAWABLEs`. It works like a movie player for a movie, where the player is the `ESDL_SPRITE` and the movie is the `ARRAY` of `ESDL_DRAWABLEs`. To use an `ESDL_SPRITE` you first create an `ARRAY` of `ESDL_DRAWABLEs` and pass it to an `ESDL_SPRITE` with *make* (*an\_animation: ARRAY [ESDL\_DRAWABLE]*). You can start and stop the playback with the feature *set\_stop* (*b: BOOLEAN*). You can set the playback speed with the feature *set\_frame\_rate* (*i: INTEGER*). The *frame\_rate* indicates how many different pictures of the animation are shown per second. The *frame\_rate* can be less than zero, the animation is then played in reverse order. The *frame\_rate* **can not** be zero, use *set\_stop* (*true*) instead. If *repeat* is true the animation is played over and over again, otherwise the playback ends at the last frame.

### 2.4.1 ESDL\_ANIMATION

An `ESDL_ANIMATION` is an `ARRAY` of `ESDL_DRAWABLEs`.

An ESDL\_ANIMATION can be played by an ESDL\_SPRITE.

Besides the normal ARRAY creations: *make* (*min\_index*, *max\_index*: *INTEGER*) and *make\_from\_array* (*a*: *ARRAY* [*ESDL\_DRAWABLE*]) you can also create an ESDL\_ANIMATION with *make\_from\_file* (*file\_name*: *STRING*) where *file\_name* is the name of a file describing the animation. The format of the animation file is a list of file names prefixed by the key color. More formal in EBNF (where nl means new line):

```
File      ::= Color_key File_list eof
Color_key ::= color_key_red nl color_key_green nl color_key_blue
File_list ::= [nl file_name_of_picture]*
```

Each *file\_name\_of\_picture* is the file name of a picture like a BMP or GIF file. An example file *f1.anim* from the ESDL Racing example on page 29 looks like the animation script shown below.

```
0
0
0
./pics/f11.gif
./pics/f12.gif
./pics/f11.gif
./pics/f13.gif
```

### 2.4.2 ESDL\_RANDOM\_SPRITE

An ESDL\_RANDOM\_SPRITE is an ESDL\_SPRITE where the frames are not shown in order but a randomly chosen frame is shown. If *respect\_speed* is *true* then a new frame is chosen each  $\frac{1}{\text{frame\_rate}}$  seconds, otherwise an other frame is chosen each call of *draw* and *draw\_part*.

### 2.4.3 Example

```
local
  anim: ESDL_ANIMATION
  sprite: ESDL_SPRITE
do
  create anim.make_from_file ("f1.anim")
  create sprite.make (anim)
  sprite.draw (screen)
  screen.redraw
end
```

## 2.5 ESDL\_TILE\_PATTERN

An ESDL\_TILE\_PATTERN is an ESDL\_DRAWABLE.

ESDL\_TILE\_PATTERN is deferred. The two children ESDL\_TILE\_PATTERN\_HORIZONTAL and ESDL\_TILE\_PATTERN\_VERTICAL make ESDL\_TILE\_PATTERN effective.

In an ESDL\_TILE\_PATTERN an ESDL\_DRAWABLE works like a tile that is glued together a number of times in either horizontal or vertical direction to build a pattern. The feature *set\_times* (*an\_integer: INTEGER*) allows to define how many times the ESDL\_DRAWABLE is repeated to build the pattern.

### 2.5.1 ESDL\_TILE\_PATTERN\_HORIZONTAL

An ESDL\_TILE\_PATTERN\_HORIZONTAL is an ESDL\_TILE\_PATTERN.

In an ESDL\_TILE\_PATTERN\_HORIZONTAL an ESDL\_DRAWABLE is glued together in horizontal direction.

### 2.5.2 ESDL\_TILE\_PATTERN\_VERTICAL

An ESDL\_TILE\_PATTERN\_VERTICAL is an ESDL\_TILE\_PATTERN.

In an ESDL\_TILE\_PATTERN\_VERTICAL an ESDL\_DRAWABLE is glued together in vertical direction.

### 2.5.3 Example

```
local
  picture: ESDL_SURFACE
  tph: ESDL_TILE_PATTERN_HORIZONTAL
do
  create picture.make_from_image ("lovely.gif")
  create tph.make (picture, 3)
  tph.draw (screen)
  screen.redraw
end
```

## 2.6 ESDL\_NEVER\_ENDING\_BACKGROUND

An ESDL\_NEVER\_ENDING\_BACKGROUND is an ESDL\_DRAWABLE.

ESDL\_NEVER\_ENDING\_BACKGROUND is deferred. The two children ESDL\_NEVER\_ENDING\_BACKGROUND\_HORIZONTAL and ESDL\_NEVER\_ENDING\_BACKGROUND\_VERTICAL make ESDL\_NEVER\_ENDING\_BACKGROUND effective.

You can think of an ESDL\_NEVER\_ENDING\_BACKGROUND as the kind of background used in old movies if a scene in a car is recorded. It is a large picture that is pulled behind the car. An ESDL\_NEVER\_ENDING\_BACKGROUND is very similar to that concept too. It expects an ESDL\_DRAWABLE as argument. This ESDL\_DRAWABLE is glued together over and over again, either in




Figure 2: A never ending background

horizontal or vertical direction, to fill the whole size of the screen. Internally `ESDL_NEVER_ENDING_BACKGROUND` uses a `ESDL_TILE_PATTERN`.

You can set the speed (in pixels per second) of the background with the command `set_speed`. The background is then moved in either horizontal or vertical direction. See figure 2.

### 2.6.1 `ESDL_NEVER_ENDING_BACKGROUND_HORIZONTAL`

An `ESDL_NEVER_ENDING_BACKGROUND_HORIZONTAL` is an `ESDL_NEVER_ENDING_BACKGROUND` constraint to horizontal movement. A positive *speed* value moves the background from right to left. A negative one in the other direction.

### 2.6.2 `ESDL_NEVER_ENDING_BACKGROUND_VERTICAL`

An `ESDL_NEVER_ENDING_BACKGROUND_VERTICAL` is an `ESDL_NEVER_ENDING_BACKGROUND` constraint to vertical movement. A positive *speed* value moves the background from below to top. A negative one in the other direction.

### 2.6.3 Example

```
local
  picture: ESDL_SURFACE
  background:
    ESDL_NEVER_ENDING_BACKGROUND_HORIZONTAL
do
  create picture.make_from_image ("lovely.gif")
  create background.make (picture)
  background.set_speed (10)
  background.draw (screen)
  screen.redraw
end
```

## 2.7 ESDL\_DRAWABLE\_CONTAINER

An ESDL\_DRAWABLE\_CONTAINER is an ESDL\_DRAWABLE and a DS\_LINKED\_LIST of ESDL\_DRAWABLEs.

If an ESDL\_DRAWABLE\_CONTAINER is drawn to an ESDL\_SURFACE all its elements are drawn to this ESDL\_SURFACE. First element of the DS\_LINKED\_LIST is drawn first, last is drawn last.

The ESDL\_DRAWABLE\_CONTAINER is a very powerful class since an ESDL\_DRAWABLE\_CONTAINER represents a new coordinate system for all its elements. That means, that the new coordinate origin is the top left pixel of the ESDL\_DRAWABLE\_CONTAINER. For example: You add an ESDL\_SURFACE with position (10,10) to an ESDL\_DRAWABLE\_CONTAINER at position (20,20). If you then draw the ESDL\_DRAWABLE\_CONTAINER to the screen, the ESDL\_SURFACE is drawn to position (30,30) at the screen. See figure 3 on the next page.




Figure 3: A container

The ability to create new coordinate systems allows to build tree structures of ESDL\_DRAWABLEs. You can build ESDL\_DRAWABLE\_CONTAINERS and add these containers to other ESDL\_DRAWABLE\_CONTAINERS and so on (Composite Pattern, see [2] and [7]).

If you need to iterate over the elements of the container you have to create your own DS\_LINKED\_LIST\_CURSOR. The reason for that is, if you compile your project in Eiffel Studio, not all your code seems to run in one thread, but in finalized programs all code does. Since we want to be able to debug our code, we have to make sure that the compiled code behaves the same way as the finalized code. Therefore every iteration has to be done with a separate cursor.

### 2.7.1 Example

```
local
  a_container:
    ESDL_DRAWABLE_CONTAINER [ESDL_SURFACE]
  alien: ESDL_SURFACE
  cursor: DS_LINKED_LIST_CURSOR [ESDL_SURFACE]
do
  create alien.make.from.image ("alien.gif")
```

```
alien.set_x_y (10, 10)
create a_container.make
a_container.set_x_y (20, 20)
a_container.extend (alien)
a_container.draw (screen)
screen.redraw
create cursor.make (a_container)
from
    cursor.start
until
    cursor.off
loop
    cursor.item.set_x (cursor.item.x + 10)
    cursor.forth
end
end
```

## 2.8 ESDL\_STRING

An ESDL\_STRING is an ESDL\_DRAWABLE.

An ESDL\_STRING expects a STRING and an ESDL\_FONT at creation. ESDL\_STRING uses an ESDL\_DRAWABLE\_CONTAINER containing a list of ESDL\_CHARACTERS. The ESDL\_FONT stored in *font* is used to draw every character of the STRING stored in *value*. ESDL\_STRING is mutable, that means you can change the *value* with *set\_value* (*a\_value*: *STRING*). This is a slow operation. You can also change the used ESDL\_FONT with *set\_font* (*a\_font*: *ESDL\_FONT*).

### 2.8.1 ESDL\_CHARACTER

An ESDL\_CHARACTER is an ESDL\_DRAWABLE.

An ESDL\_CHARACTER expects a CHARACTER and an ESDL\_FONT at creation. The ESDL\_FONT stored in *font* is used to draw the CHARACTER stored in *character*.

### 2.8.2 ESDL\_FONT

The deferred class ESDL\_FONT should be implemented by every kind of font. An ESDL\_FONT can draw a CHARACTER or part of it to an ESDL\_SURFACE to a given position. An ESDL\_FONT can also be asked for the *width* and the *height* of a CHARACTER.



```
deferred class
  ESDL_FONT

feature -- commands

  draw (a_character: CHARACTER;
        a_surface: ESDL_SURFACE;
        x: INTEGER; y: INTEGER) is
    -- Draws 'a_character' to 'a_surface'
    -- to position 'x' 'y'
  require
    a_surface /= void
    a_character /= void
  deferred
  end

  draw_part (rect: ESDL_RECT; a_character: CHARACTER;
            a_surface: ESDL_SURFACE;
            x: INTEGER; y: INTEGER) is
    -- Draws 'rect' part of 'a_character' to
    -- 'a_surface' to position 'x' 'y'
  require
    a_surface /= void
    a_character /= void
    rect /= void
  deferred
  end

feature -- queries

  width (a_character: CHARACTER): INTEGER is
    -- The width of 'a_character'
  deferred
  end

  height (a_character: CHARACTER): INTEGER is
    -- The height of 'a_character'
  deferred
  end

end
```




Figure 5: Bitmap font Arial\_256

At the moment only one class is effecting ESDL\_FONT: The ESDL\_BITMAPFONT. In the future an ESDL\_VECTORFONT can be implemented without changing any class. See figure 4.




Figure 4: ESDL\_FONT BON Diagram

### 2.8.3 ESDL\_BITMAPFONT

ESDL\_BITMAPFONT is an ESDL\_FONT.

An ESDL\_BITMAPFONT expects an ESDL\_DRAWABLE that represents a matrix. Every row and every column contains 16 ASCII characters. That are  $16 \times 16 = 256$  characters. The character with the ASCII number 0 is at the top left, the character with the ASCII number 255 is at the bottom right. The program Bitmap font Builder ([www.lmnop.com/bitmapfontbuilder/](http://www.lmnop.com/bitmapfontbuilder/)) can create such images (see figure 5)

The *height* of a CHARACTER is the *height* of *font* divided by 16, the *width* is the *width* of *font* divided by 16.

Since you can pass an ESDL\_DRAWABLE as matrix you can, for example, build animated fonts. You should not pass an ESDL\_NEVER\_ENDING\_BACKGROUND because of the resulting *width* or *height* of a CHARACTER.

#### 2.8.4 Example

```
local
  str: ESDL_STRING
  font: ESDL_BITMAPFONT
  matrix: ESDL_SURFACE
do
  create matrix.make_from_image ("arial256.gif")
  create font.make (matrix)
  create str.make ("Hello world", font)
  str.draw (screen)
  screen.redraw
end
```

### 2.9 ESDL\_FRAME\_COUNTER

An ESDL\_FRAME\_COUNTER is an ESDL\_DRAWABLE.

An ESDL\_FRAME\_COUNTER is a very useful utility. You can *draw* a frame counter every frame to the screen and get information about the performance of your program. The frame counter is used in many of the examples. It calculates the frame rate averaged over the last 10 frames. The frame rate is a measure for how many times per second the computer executes *draw* or *draw\_part* feature of ESDL\_FRAME\_COUNTER. ESDL\_FRAME\_COUNTER uses an ESDL\_BITMAPFONT to print the rate to *a\_surface*. If you use an ESDL\_FRAME\_COUNTER you have to provide a copy of the file *arial256.gif* in the same directory where the ACE file of your program is. You find the file *arial256.gif* in most of the example directories (see figure 5 on the preceding page).

#### 2.9.1 Example

```
local
  fc: ESDL_FRAME_COUNTER
do
  create fc.make
  fc.draw (screen)
  screen.redraw
end
```

### 2.10 ESDL\_TIME

The class ESDL\_TIME provides the ESDL programmer with functionalities related to time measurement. The two most important features are *delay* (*mil-*

*liseconds*: *INTEGER*) and *get\_ticks*: *INTEGER*. A call to the command *delay* needs at least *milliseconds* time. This gives you the possibility to hold the execution of your program for a given amount of time. The SDL documentation states that:

Note: Count on a delay granularity of at least 10 ms. Some platforms have shorter clock ticks but this is the most common.

The feature *get\_ticks* returns the number of milliseconds since ESDL initialization. Do not count on getting precise results from *get\_ticks* either. If you need very precise information you have to calculate an average somehow. I did that in the feature *update\_cur\_position* of the class *ESDL\_NEVER\_ENDING\_BACKGROUND*, to allow very smooth motion of a never ending background. Since many of the classes in ESDL require that the time subsystem is initialized, the time subsystem is initialized when you create an ESDL object. You can not instantiate the class *ESDL\_TIME* afterwards, if you need its functionality inherit from *ESDL\_TIME\_SINGLETON* and call feature *get\_time*: *ESDL\_TIME*.

### 2.10.1 Example

```
inherit
  ESDL_TIME_SINGLETON

feature

  test is
    local
      i: INTEGER
    do
      i := get_time.get_ticks
      get_time.delay (1000)
      i := get_time.get_ticks - i
      print (i.out + " >= 1000 may be true")
    end
```

### 2.10.2 ESDL\_TIME\_SINGLETON

*ESDL\_TIME\_SINGLETON* ensures that only one instance of *ESDL\_TIME* exists in the system. To get that instance inherit from *ESDL\_TIME\_SINGLETON* and call *get\_time*: *ESDL\_TIME*. Since Eiffel doesn't support explicit static called features the pattern described in [3] and implemented by *ESDL\_TIME\_SINGLETON* is a possibility to ensure that only one instance of a class exists in the system, or at least, make it hard to build a second one. The reason why *ESDL\_TIME* must be a singleton is its ability to store and call a list of callbacks.

### 2.10.3 Callbacks

An other feature of ESDL\_TIME is the feature *add\_timed\_callback* (*interval*: *INTEGER*; *an\_action*: *FUNCTION* [*ANY*, *TUPLE* [*INTEGER*], *INTEGER*])

The feature allows to pass a pointer to a function (agent) which has one parameter of type *INTEGER* and an *INTEGER* as result. This function is called after *interval* milliseconds. The value of the parameter is either the number of milliseconds since the function was called last time or interval if it is the first call. The resulting value can be less or equal to 0 if you don't want ESDL\_TIME to call the function again or any number greater 0. If the number is greater 0 the function is called again after *result* number of milliseconds. The agent *an\_action* should always be executed faster than *interval*. The agent *an\_action* runs in a separate thread. You can use ESDL\_MUTEX to synchronize between two threads.

### 2.10.4 Example

```
make is
do
  get_time.add_timed_callback
  (100, agent action (?))
end

action (interval: INTEGER): INTEGER is
  -- action is called every 100 milliseconds
do
  Result := interval
end
```

### 2.10.5 ESDL\_MUTEX

You can use an ESDL\_MUTEX to synchronize threads. Its two features are *lock* and *unlock*. The feature *lock* waits until the caller gets a lock on current. The feature *unlock* releases the lock.

## 2.11 ESDL\_COLLIDABLE

Classes describing objects that can collide with other objects should be of type ESDL\_COLLIDABLE. If an object collides with an other object *on\_collide* (*other*: *ESDL\_COLLIDABLE*) is called by the ESDL\_COLLISION\_DETECTOR. The *type\_id* query returns a number that is the same for all instances of a class. This allows in *on\_collide* to determine of what kind the other object is.

```
deferred class
  ESDL_COLLIDABLE

feature -- Status report

  does_collide (other: like Current): BOOLEAN is
    -- Does 'current' collide with 'other'?
    require
      other_not_void: other /= void
    deferred
    end

  type_id: INTEGER is
    -- The id given to all instances of 'current' generating
    -- class
    deferred
    end

feature {ESDL_COLLISION_DETECTOR} -- Implementation

  on_collide (other: ESDL_COLLIDABLE) is
    -- 'current' collided with 'other'
    require
      other_not_void: other /= void
    deferred
    end

end
```

### 2.11.1 ESDL\_RECT\_COLLIDABLE

ESDL\_RECT\_COLLIDABLE is an ESDL\_COLLIDABLE. An ESDL\_RECT rectangle is used to describe the size of an object. This rectangle is called *bounding\_box*. If the intersection of the *bounding\_box* of one ESDL\_RECT\_COLLIDABLE object and an other is not empty, the two objects collide. Finding out if such an intersection is empty or not, can be done very fast.

```
deferred class
  ESDL_RECT_COLLIDABLE

inherit
  ESDL_COLLIDABLE

feature -- Status

  does_collide (other: like Current): BOOLEAN is
    -- Does 'current' collide with 'other'?
  do
    Result :=
      Current.bounding_box.intersects
        (other.bounding_box)
  end

  bounding_box: ESDL_RECT is
    -- The active elements of Current
  deferred
  end

end
```

### 2.11.2 ESDL\_COLLISION\_DETECTOR

An ESDL\_COLLISION\_DETECTOR can be used to check if any of its containing ESDL\_COLLIDABLE objects collide with any other. You can add an ESDL\_COLLIDABLE to an ESDL\_COLLISION\_DETECTOR with the command *add* (*a\_collidable*: ESDL\_COLLIDABLE). Every time you call *check\_for\_collision* every object added to ESDL\_COLLISION\_DETECTOR is tested with every other object added to ESDL\_COLLISION\_DETECTOR for collision ( $O(n^2)$ ). If a collision is detected either the *on\_collide* feature of one of the two colliding objects is called if *call\_both* is equals *false* or the *on\_collide* features of both objects are called otherwise.

### 2.11.3 Example

See the *collidable\_string* example on page [28](#).

## 3 Examples

### 3.1 Bitmap fonts




Figure 6: Bitmap font example

The bitmap font example demonstrates how to use the class `ESDL_BITMAPFONT` to print characters on the screen.



### 3.2 Drawable demo




Figure 7: A drawable demo part 1



Figure 8: A drawable demo part 2

The `drawable_demo` example demonstrates how you can combine some of the `ESDL_DRAWABLE` classes to build more and more complex objects.

### 3.3 Never ending background



Figure 9: A demo of a never ending background

The `never_ending_background` example demonstrates how you can combine an `ESDL_NEVER_ENDING_BACKGROUND_HORIZONTAL` with an `ESDL_NEVER_ENDING_BACKGROUND_VERTICAL` to build a never ending background in any direction.

### 3.4 Collidable string




Figure 10: A demo of collidable strings

The `collidable_string` example demonstrates how you can use the `ESDL_RECT_COLLIDABLE` class in combination with the `ESDL_COLLISION_DETECTOR` class to detect collisions of objects.

### 3.5 ESDL Racing



Figure 11: ESDL Racing level 1




Figure 12: ESDL Racing level 2




Figure 13: ESDL Racing level 3

The `esdl.racing` example is the largest example. It is a fully playable racing game. All of the abilities of the classes described in this semester thesis report are used. The example also demonstrates how you can implement menus connecting multiple scenes. Every scene should inherit from the deferred class `ESDL_SCENE`. The class `MENU` and the class `LEVEL` both extend `ESDL_SCENE`. An `ESDL_SCENE` contains an *event\_loop* and the very important feature *next\_scene*: *ESDL\_SCENE*. After current scene ends the *next\_scene* is initialized by *initialize\_scene* and then executed by calling *run* from the main loop if *next\_scene* is not *void*:

```
-- Main loop
from
    scene := create {START}
until
    scene = void
loop
    scene.initialize_scene
    scene.run (screen)
    scene := scene.next_scene
end
```

`START` is the first scene, the main menu you see when you execute `esdl.racing`.

If you want to create your own level inherit from `LEVEL`. See one of the classes `LEVEL01`, `LEVEL02` or `LEVEL03` as an example. To make sure your

level is executed after LEVEL03 set the *next\_scene* of LEVEL03 accordingly in the feature *on\_win*. To make testing easier there exist two cheat codes: press c to turn off collision detection, press e to jump directly to the end of a level.

## 4 Future Work

- Implement scaling, rotation support and panning for surfaces  
Using a filtering approach (A deferred FILTER class, together with classes effecting it - e.g. SCALE\_FILTER that takes a surface and returns one).
- Extend event handling support for mouse events
- Extend font support  
ESDL\_VECTORFONT - eventually with the metafont project that would have to be wrapped separately.
- Implement audio subsystem  
A format independent approach like the one taken for image formats is desirable.
- Implement EiffelVision2 widgets (Linux and Windows)
- Extend video subsystem API with primitives (circles, lines, rectangles etc.), anti aliasing, alpha channels, YUV

## References

- [1] Till G. Bay, 2003, Eiffel SDL multimedia library (ESDL) Version 0.0.1  
<http://n.ethz.ch/student/bayt/publications>
- [2] GOF patterns for GUI Design, 1998, James Noble, side 6  
[citeseer.nj.nec.com/noble98gof.html](http://citeseer.nj.nec.com/noble98gof.html)
- [3] How to get a Singleton in Eiffel?, 2004, Karine Arnout, Éric Bezault  
<http://se.inf.ethz.ch/people/arnout/arnout-bezault-singleton.pdf>
- [4] Simple Directmedia Layer  
<http://www.libsdl.org>
- [5] Eiffel Wrapper Generator  
<http://ewg.sourceforge.net/>
- [6] SDL image  
[http://www.libsdl.org/projects/SDL\\_image/](http://www.libsdl.org/projects/SDL_image/)
- [7] Design Patterns, 1995, Eric Gamma, Richard Helm, Ralph Johnson, John Vlissides  
<http://www.amazon.com/exec/obidos/ASIN/0201633612/javaworld>